
C H A P T E R  1 SAMPLE S01.M01

Introduction 
This book is the written version of the Optimizing DAX video course (second edition) published by SQLBI. 
While the PDF version is available to video course students, we decided to publish a printed version for 
those who want a printed version or do not want to get the entire video course and prefer to study books. 

Several years ago, we recorded and published the first edition of the Optimizing DAX video course. At 
the time, Tabular was much simpler than it is today, and the DAX optimization was described in a few 
chapters of “The Definitive Guide to DAX” book, including some details about the VertiPaq model. In the 
first edition of that training, we did not cover – on purpose – DirectQuery optimizations. Indeed, 
DirectQuery was not a real option at that time. 

Over the years, Tabular has evolved, and its complexity is now much more significant. Microsoft 
introduced a more usable version of DirectQuery, released composite models, and improved the DAX 
engine by adding new optimization techniques. Moreover, as teachers, we increased our knowledge and 
experience about the engine in the same period. When it was time to create a new training version, we 
quickly discovered the amount of materials to teach was massive. Therefore, despite this being the second 
edition of the Optimizing DAX video course, it is – in reality – a brand new training (and book) about the 
state of the art of optimizing Tabular models. 

An important note is that we wrote this content using Analysis Services 2022 and Power BI versions 
available in 2023. Different engine versions might show different behaviors, for the better or the worse. 
We will say this for the first time now and repeat it multiple times during the training: you must test any 
of the optimization techniques we teach on your models and the version of the engine you are currently 
using. 

Prerequisites 

The content of this book is advanced. As such, it certainly does not start from scratch. We take for granted 
that the reader has an excellent knowledge of several topics: 

• The DAX language. In the training, we never teach DAX concepts. Here, the goal is how to
produce efficient DAX code. We will write a lot of DAX code together, taking it for granted that
you will quickly understand the different formulas. If you are unfamiliar with DAX, we strongly
suggest the Mastering DAX video course, “The Definitive Guide to DAX” book, and practicing the
DAX language in your daily job for at least one year. Being proficient with DAX is an absolute
requirement. The book also uses query columns and query tables described in these articles:

• Introducing DEFINE COLUMN in DAX queries
https://www.sqlbi.com/articles/introducing-define-column-in-dax-queries/

https://www.sqlbi.com/articles/introducing-define-column-in-dax-queries/

• Introducing DEFINE TABLE in DAX queries
https://www.sqlbi.com/articles/introducing-define-table-in-dax-queries/

• The SQL language. The modules about DirectQuery use quite some SQL code. You do not need
to master all the details about SQL. Still, you must be able to read and understand SQL code
quickly and how indexes, column stores, and other relational database technologies may affect
your query’s results and performance. We use Microsoft SQL Server in the book examples. You
can apply the same concepts to other relational databases – even though you cannot use the same 
specific techniques described for Microsoft SQL Server.

• The Tabular architecture. Tabular is a complex engine that includes different technologies. We
will use but do not explain many features of Tabular. Even though you can read and learn most of
the content of this book without a deep understanding of the Tabular engine, we strongly suggest
our students attend the Mastering Tabular video course first. Unfortunately, we do not have a
corresponding book for that content.

• DAX Studio. DAX Studio is the primary tool used throughout the entire training. We introduce
some basic concepts about DAX Studio but do not go in-depth on all the tool features. If you are
unfamiliar with DAX Studio, we suggest you attend the free DAX Tools training first at
https://www.sqlbi.com/p/dax-tools-video-course/.

Overview of the Tabular architecture 

The architecture of a modern Tabular solution can be rather complex. To perform any optimization, we 
must understand many technical details about a Tabular solution. Indeed, when there are performance 
issues, users typically report that “the dashboard is slow” or “it takes too long to produce the report”. 
However, the report is only the tip of the iceberg of a performance issue hidden in any place of the 
architecture. 

The following figure shows a very high-level overview of what happens when rendering a report: 

https://www.sqlbi.com/articles/introducing-define-table-in-dax-queries/
https://www.sqlbi.com/p/dax-tools-video-course/

The report sends a query to the Tabular model, which runs the query and might need to access an SQL 

database if the model uses DirectQuery, scan a VertiPaq storage if the model is in import mode, or connect 
to another Tabular model if we are using a composite model. Worse, it might be a mix of all these 
technologies together in a complex model. 

The optimization techniques to use strongly depend on the architecture. We might face a simple DAX 
issue: poor code leads to bad performance. However, the same DAX code might work well with VertiPaq, 
and it might be a severe issue in a composite model. It might be the case that the code itself works fine, 
but the amount of materialization is excessive because the model uses DirectQuery. 

In other words, providing guidelines about authoring good DAX code and building a performant 
model is nearly impossible unless we clearly understand all the pieces connected in a Tabular model. 

Structure of the training 

The book has five sections: 

• Section 1: Core concepts. This introductory section starts with some optimization examples 
to provide an overview of the steps needed to optimize DAX code or Tabular models. Then, 
it covers the main topics about the Tabular architecture, includes information about the tools 
to use, and introduces the query plan concept. 

• Section 2: The formula engine. This section is no longer an introduction. We go as deep as 
possible into the details of the formula engine, discover its operators, understand the 
datacaches, read the query plan, and then start optimizing code. 

• Section 3: VertiPaq. Time for the main course. VertiPaq is the most critical section. We dive 
into the details of the VertiPaq engine through seven dense chapters packed with technical 

information. The number of scenarios we optimize increases significantly, along with our 
knowledge. 

• Section 4: DirectQuery over SQL. This section explores all the details about the DirectQuery 
over SQL storage engine. Please do not skip the previous sections of the book; they are 
required to understand how DirectQuery works. 

• Section 5: Composite models. Composite models are a recent architectural addition to 
Tabular, and they prove to be more challenging because they share all the complexity of 
VertiPaq and DirectQuery. 

A first read of the book from cover to cover reveals a long journey in DAX optimizations. Once you 
finish the book, you can use it as a reference to refresh your mind about specific topics. 

We advise readers not to jump directly to the more advanced sessions until they are familiar with the 
previous topics. We build knowledge in steps and never repeat a topic multiple times. If you fast-forward 
to an advanced chapter, it is unlikely you will be able to appreciate several of the nuances. 

Coding conventions 

The book has many code samples in three languages: DAX, SQL, and xmSQL. You will learn xmSQL in this 
book, whereas you should already know DAX and SQL. Every language has a different style for code 
snippets that we present here to familiarize ourselves with. 

Here is a sample code in DAX: 

EVALUATE 
SUMMARIZECOLUMNS ( 

 ROLLUPADDISSUBTOTAL ( 
 'Date'[Year], 
"IsGrandTotalRowTotal", 
ROLLUPGROUP ( 'Date'[Month], 'Date'[Month Number] ), 
"IsDM1Total" 

 ), 
    "Sales_Amount", 'Sales'[Sales Amount] 
) 

The following is SQL: 

SELECT TOP (1000001) * 
FROM ( 
    SELECT [t1_Year], SUM([a0]) AS [a0] 
    FROM ( 

SELECT [t1].[Year] AS [t1_Year], 
([t3].[Quantity] * [t3].[Net Price]) AS [a0] 

FROM ( 
[Data].[Sales] AS [t3] 

LEFT JOIN [dbo].[Date] AS [t1] ON ([t3].[Order Date] = [t1].[Date]) 
) 

) AS [t0] 
    GROUP BY [t1_Year] 
    ) AS [MainTable] 
WHERE (NOT (([a0] IS NULL))) 

And the last example is xmSQL: 

WITH 
  $Expr0 := ( PFCAST ( 'Sales'[Quantity] AS  INT ) * PFCAST ( 'Sales'[Net Price] AS  INT )  ) 

SELECT 
  'Product'[Brand], 
  SUM ( @$Expr0 ) 

FROM 'Sales' 
  LEFT OUTER JOIN 'Product' 

 ON 'Sales'[ProductKey]='Product'[ProductKey]; 



8 CHAPTER 1    Introduction    S01.M01 

Companion content 

All the examples included in the book can be downloaded and reproduced on your computer. You might 
see different absolute results because of differences in CPU, RAM, and software versions. However, most 
of the time, you will see the same relative differences between different optimization steps. Keep in mind 
the different baselines between your computers and ours. Because we used different hardware in different 
parts of the book, you will always find a reference to the CPU used in the initial comments of the DAX 
queries you can download. 

You can download all the files following the link to Companion content on the 
https://sql.bi/optdaxdemo page.  

In the following sections, we provide software and hardware prerequisites to use the demo files, how 
to download them, and how to find the demo files corresponding to each book chapter. 

Software prerequisites 
You should have the following software to practice using the book demos. If you want to know more 
about the optional tools, you can find more information in the Mastering Tabular video course from SQLBI 
at https://www.sqlbi.com/p/mastering-tabular-video-course/. 

• Power BI Desktop 
https://powerbi.microsoft.com/downloads/ 

• DAX Studio 
https://daxstudio.org/ 

• Excel (optional) 
https://www.microsoft.com/microsoft-365/excel 

• SQL Server 2022 (optional) 
https://www.microsoft.com/sql-server/ 

• SQL Server 2022 latest cumulative updates (optional) 
https://www.microsoft.com/en-us/download/details.aspx?id=105013 

• SQL Server Analysis Services Tabular (optional) 
https://learn.microsoft.com/analysis-services/tabular-models/tabular-models-ssas 

• SQL Server Management Studio (optional) 
https://learn.microsoft.com/sql/ssms/download-sql-server-management-studio-ssms 

• Tabular Editor 2 or 3 (optional) 
https://www.tabulareditor.com/ 

https://sql.bi/optdaxdemo
https://www.sqlbi.com/p/mastering-tabular-video-course/
https://powerbi.microsoft.com/downloads/
https://daxstudio.org/
https://www.microsoft.com/microsoft-365/excel
https://www.microsoft.com/sql-server/
https://www.microsoft.com/en-us/download/details.aspx?id=105013
https://learn.microsoft.com/analysis-services/tabular-models/tabular-models-ssas
https://learn.microsoft.com/sql/ssms/download-sql-server-management-studio-ssms
https://www.tabulareditor.com/

Hardware prerequisites 
To practice the demos with the largest PBIX demo file (Contoso 100M.pbix, 2GB), you need a computer 
with at least 16GB of RAM. If you do not have enough RAM, you can use the smaller Contoso 10M.pbix, 
which should run on any computer running Power BI Desktop, even though the metrics obtained can be 
significantly different from the ones shown in the book. 

If you want to create the largest database for Analysis Services (Contoso 100M x 10), we suggest a 
computer with at least 64GB of RAM. This model is used in a few demos: you do not need to repeat those 
same demos, as you can see similar effects on smaller models, even though the results will be at a different 
order of magnitude. 

Download demos 
The demos are files we use in the book to illustrate how to measure performance and optimize code. You 
need Power BI Desktop and DAX Studio to repeat the same demos on your PC. For a few demos that 
require very large databases, you need SQL Server and Analysis Services: if you do not have enough RAM 
to install those databases, repeat the demos on the smaller Power BI Desktop files. 

To practice the concepts learned, we suggest looking for patterns you have seen in each section by 
repeating the same analyses with the sample demo files provided and then on your models. 

All the Contoso sample databases have the same structure; the only difference is their size. For 
example, Contoso 10K contains around 10,000 orders, while Contoso 1M contains around 1 million orders. 
Most of the demos run on Contoso 100M, and few run on a larger or smaller Contoso database: you find 
the reference to the Contoso version in the comments at the beginning of each DAX demo file. 

The PBIX files contain the sample database for Power BI Desktop. You do not need to refresh these 
databases. Still, if you want, you should download the SQL Server backup (.bak files), restore them with 
the same database name (without the .bak extension) on a local SQL Server database, and create an alias 
named Demo that points to the SQL Server instance where you restored the bak files. Read the Creating 
database aliases on SQL Server 2022 article to create the Demo aliases on both 32-bit and 64-bit 
configurations of SQL Server: https://sql.bi/creating-aliases/. 

The Optimizing DAX - Demo files.ZIP archive contains all the files used in the demos of the book. 
Download the file and unzip the content in a local folder on your computer. Most of the demos run on a 
Contoso sample database; only a limited number of demos have separated PBIX files that you find in the 
same module folder of the demo files. 

The BAK files are SQL Server backups you can restore on SQL Server 2019 or 2022 to refresh the PBIX 
file and/or to populate corresponding Analysis Services databases. Read the Creating database aliases on 
SQL Server 2022 article to create the Demo aliases on both 32-bit and 64-bit configurations of SQL Server: 
https://sql.bi/creating-aliases/. The Contoso 100M.bak file is stored in two files, Contoso 100M.7z.001 and 
Contoso100M.7z.002: you can extract the Contoso 100M.bak using 7-Zip (https://www.7-zip.org/). 

The AS BIM - Contoso 100M X 10.ZIP archive contains the Tabular project (BIM file) to generate the 

https://sql.bi/creating-aliases/
https://sql.bi/creating-aliases/
https://www.7-zip.org/

Analysis Services database "Contoso 100M x 10" used with the demos with the largest database. You need 
a large server with at least 64GB of RAM to process this database. However, you can use the Contoso 
100M PBIX file to run the same demos executed on "Contoso 100M x 10" (see the reference database in 
the comments of the DAX code): you will see similar effects at a smaller order of magnitude. 

Sample code references 
At the beginning of each book chapter, there is a reference to the sample code. This reference is the 
corresponding section and module of the Optimizing DAX video course on SQLBI. For example, SAMPLE 
S02.M03 means that the sample code for the chapter is in the folder starting with "S02.M03" within the 
Demo folder. Such a reference also corresponds to Section 02 and Module 03 of the video course. 

C H A P T E R  2 SAMPLE S01.M02

Introducing optimization with 
examples 

The amount of knowledge required to optimize any measure is massive. The architecture of Tabular is 
rather complex, and we will dive into extremely technical details to appreciate the effect of any change. 

It is easy to lose sight of the overall reason why certain specifics are important when diving into details. 
Therefore, if we were to follow an academic approach, and explain all the details before starting to use 
the concepts, the training would be extremely hard to follow, somewhat boring, and you would have to 
constantly go back to previous chapters to refresh your memory about topics that did not seem relevant 
the first time around. 

We chose a different approach. This first set of introductory content shows the complete process of 
optimization through examples. We neither explain each step in full nor do we explain why we perform 
certain actions. The goal of this first set of optimizations is not that of learning how to optimize. Instead, 
we want to show the reasoning required to perform optimizations in different architectures. 

Do not consider these first optimizations as best practices, and do not try to derive complete 
knowledge from these examples. These are just simple examples to show the overall process. Later on, 
we shall start to dive into the details. Then, hopefully, having seen the complete process will help you 
focus on the details without losing sight of the big picture. 

During the process of optimizing the code, we also use several tools. Again, we do not provide a 
complete reference of those tools, because the goal is to show you when the tools are used rather than 
explain how to use them. We provide a much greater level of detail in the following chapters. In the entire 
description, we are deliberately concise. If the description seems cloudy to you, this is to be expected. Our 
promise is that you will be able to perform the same operations once you complete the training and have 
some more experience under your belt.  

Optimizing DAX 

We start by analyzing a simple measure that contains a DAX issue. The code is poorly written, and in the 
process of trying to optimize it, we will also make it worse before finally reaching the optimized version. 
The example is not relevant, despite being quite common. The critical detail to focus on is the entire 
optimization process. 

It all starts with a report that users describe as slow. It contains a simple measure that computes sales 
for only transactions whose amount is greater than 200.00 USD. 

The code of the measure is the following: 

--------------------------- 
-- Measure in Sales table 
--------------------------- 
Sales Gt 200 = 
SUMX ( 

 Sales, 
    IF ( 

 Sales[Quantity] * Sales[Net Price] >= 200, 
Sales[Quantity] * Sales[Net Price] 

 ) 
) 

If you are a seasoned DAX developer, you know where the problem is right off the bat. Nonetheless, 
in this section we work on the full performance analysis of the measure. 

The report is on a Power BI Desktop file, using Import mode. The Sales table contains 200M rows, 
spanning around 10 years of data. We use the Performance Analyzer tool in Power BI Desktop to retrieve 
the DAX query executed for the visual. Before analyzing the query, Performance Analyzer already provides 
us with some rather interesting numbers: the query took 1.2 seconds to run on a quite powerful server 
with 64 virtual cores. 

The code of the query is verbose, because it is automatically generated by Power BI: 

// DAX Query 
DEFINE 

 VAR __DM3FilterTable = 
TREATAS ( { 2010 }, 'Date'[Year] ) 

 VAR __DS0Core = 
SUMMARIZECOLUMNS ( 

 ROLLUPADDISSUBTOTAL ( 
 'Date'[Year], 
"IsGrandTotalRowTotal", 
ROLLUPGROUP ( 'Date'[Month], 'Date'[Month Number] ), 
"IsDM1Total", 
NONVISUAL ( __DM3FilterTable ) 

 ), 
"Sales_Gt_200", 'Sales'[Sales Gt 200] 

 ) 
 VAR __DS0PrimaryWindowed = 

TOPN ( 
 502, 
__DS0Core, 
[IsGrandTotalRowTotal], 0, 
'Date'[Year], 1, 
[IsDM1Total], 0, 
 'Date'[Month Number], 1, 
'Date'[Month], 1 

 ) 

EVALUATE 
__DS0PrimaryWindowed 
ORDER BY 

 [IsGrandTotalRowTotal] DESC, 
    'Date'[Year], 
    [IsDM1Total] DESC, 
    'Date'[Month Number], 
    'Date'[Month] 

We want to simplify the query, to make it easier to understand. In the process of simplifying the query, 

we need to pay attention not to get rid of the problem. The first thing to do is to execute the query in 
DAX Studio with Server Timings enabled to obtain the first baseline. Later on, we will check that the 
simplified query did not change the timings in such a way that the issue seems resolved. Here is the DAX 
Studio timings report. 

Then, we simplify the query by removing the TOPN function, the final sorting, and other lines to make 
it shorter. We also add the definition of the measure, so we can change it later. Here is the shorter version 
we are going to work with: 

DEFINE 
 MEASURE Sales[Sales Gt 200] = 

SUMX ( 
 Sales, 
IF ( 

 Sales[Quantity] * Sales[Net Price] >= 200, 
Sales[Quantity] * Sales[Net Price] 

 ) 
 ) 

EVALUATE 
SUMMARIZECOLUMNS ( 

 ROLLUPADDISSUBTOTAL ( 
 'Date'[Year], 
"IsGrandTotalRowTotal", 
ROLLUPGROUP ( 'Date'[Month], 'Date'[Month Number] ), 
"IsDM1Total" 

 ), 
    "Sales_Gt_200", 'Sales'[Sales Gt 200] 
) 

The timings of this query are close to those of the previous one. 

Since the numbers are similar, we know that the problem is still there and we can start the optimization 
process. Despite the size of the Sales table being relatively large (200M is not huge, but it already is a 
significant number), the time required to compute the result seems excessive. The degree of parallelism 
is exceptional (on 64 cores, we obtained x51.0). Though the entire execution time is reported as storage 
engine CPU, we can clearly see a CallbackDataID, indicating that the formula engine is required to kick in 
to compute expressions that cannot be pushed down to the storage engine. 

We know that a CallbackDataID is often a source of performance issues. It is nearly impossible to 
remove all CallbackDataIDs from a query, but we can remove it in this situation. The problem is the IF 
statement inside the iteration carried on by SUMX, because the VertiPaq storage engine does not support 
conditional logic. We must rephrase the measure to avoid the IF statement; we replace it with a condition 
set by CALCULATE to rely on filtering rather than IF. A first (wrong) attempt in this direction is the 
following: 

DEFINE 
 MEASURE Sales[Sales Gt 200] = 

CALCULATE ( 
 SUMX ( Sales, Sales[Quantity] * Sales[Net Price] ), 
FILTER ( Sales, Sales[Quantity] * Sales[Net Price] >= 200 ) 

 ) 

EVALUATE 
SUMMARIZECOLUMNS ( 

 ROLLUPADDISSUBTOTAL ( 
 'Date'[Year], 
"IsGrandTotalRowTotal", 
ROLLUPGROUP ( 'Date'[Month], 'Date'[Month Number] ), 
"IsDM1Total" 

 ), 
    "Sales_Gt_200", 'Sales'[Sales Gt 200] 
) 

The idea is to remove the requirement to compute IF by replacing it with a table filter computed by 
FILTER and applied by CALCULATE. It turns out to be an awful idea: the server timings are much worse 

than before. 

The storage engine CPU is a bit lower than the previous version of the measure, but the degree of 
parallelism is much lower this time (x6.0). Moreover, the formula engine executes a significant portion of 
code, making the overall performance much worse than the previous one. Overall, the execution time 
went from 1.5 to 23 seconds. 

A deeper analysis of the xmSQL queries shows that the storage engine is not actually computing the 
result. Although we have removed the CallbackDataID and replaced it with an xmSQL filter, the VertiPaq 
engine retrieves way too much data. The following is the first xmSQL query in the previous screenshot: 

WITH 
  $Expr0 := ( PFCAST ( 'Sales'[Quantity] AS INT ) * PFCAST ( 'Sales'[Net Price] AS INT ) ) 

SELECT 
  'Sales'[Quantity], 
  'Sales'[Net Price], 
  'Date'[Year], 
  'Date'[Month], 
  'Date'[Month Number], 
  SUM ( @$Expr0 ) 

FROM 'Sales' 
  LEFT OUTER JOIN 'Date' 

 ON 'Sales'[Order Date]='Date'[Date] 
WHERE 

 ( COALESCE ( ( PFCAST ( 'Sales'[Quantity] AS INT ) * PFCAST ( 'Sales'[Net Price] AS INT ) ) ) >= COALESCE ( 2000000 ) ) ; 

Even though the query groups by Date[Year], Date[Year Month] (and Date[Year Month Number] 
because of the sort-by-column property), the xmSQL query also groups by Sales[Quantity] and Sales[Net 

Price]. The reduced degree of parallelism is mainly due to the large size of the datacaches returned by the 
storage engine to the formula engine. Moreover, the formula engine must carry out the calculation 
because the VertiPaq result in the datacache does not contain the result. Hence the exaggerated time 
required from the formula engine. The problem is the large datacache size, resulting in extreme 
materialization. The technique in itself is smart. The problem is the way we expressed the query. 

Indeed, we used a filter over Sales as a filter argument in CALCULATE. A table filter is a very bad practice 
that newbies oftentimes use. A seasoned DAX developer knows that a filter in CALCULATE should work 
on the minimum number of columns required to obtain its effect. The filter over Sales in CALCULATE is 
for sure the issue in this measure. Therefore, we move forward and replace it with a filter over the only 
two columns required to apply their effect, which are Sales[Quantity] and Sales[Net Price]: 

DEFINE 
 MEASURE Sales[Sales Gt 200] = 

CALCULATE ( 
 SUMX ( Sales, Sales[Quantity] * Sales[Net Price] ), 
FILTER ( 

 ALL ( Sales[Quantity], Sales[Net Price] ), 
 Sales[Quantity] * Sales[Net Price] >= 200 

 ) 
 ) 

EVALUATE 
SUMMARIZECOLUMNS ( 

 ROLLUPADDISSUBTOTAL ( 
 'Date'[Year], 
"IsGrandTotalRowTotal", 
ROLLUPGROUP ( 'Date'[Month], 'Date'[Month Number] ), 
"IsDM1Total" 

 ), 
    "Sales_Gt_200", 'Sales'[Sales Gt 200] 
) 

The result is exactly what we were searching for. 

All the indicators are just perfect. The storage engine CPU is massively reduced, the degree of 
parallelism is back to being exceptional, there is virtually no formula engine involved in the query and no 
CallbackDataIDs anywhere. Materialization is reduced from 5 million rows to only 3,150 rows. Overall, the 
execution went from 1.5 seconds to a bit less than 200 hundreds milliseconds – which given the size of 
the Sales table, was expected. 

Job done; the code is now good enough to go into production and replace the previous measure. In 
order to complete our task we had to leverage on our DAX knowledge to rephrase the code, we had to 
discover what is being computed by the formula engine and the storage engine, we had to check that the 
degree of parallelism was the one expected, and we used the size of the datacaches as an indicator of the 
materialization level. We used Performance Analyzer and DAX Studio to obtain our goal. We will learn all 
these details. For now, let us move on to the next example. 

Optimizing the model 

In the previous example, we optimized a piece of DAX code: The goal was to reduce the execution time 
of a query. We started from a measure because it is likely to be the most common optimization 
requirement. Nonetheless, other types of optimization are equally important. For example, reducing the 
size of a model improves both the execution time and the memory usage of a data model. 

As in the previous example, we are not interested in providing detailed information about our 
considerations to optimize a model. Rather, we want to share the steps of optimizing a model by showing 
the reasoning behind a choice. In this example, we must choose between a calculated column and a 
calculation at query time. 

The Sales table contains two columns – Sales[Quantity] and Sales[Net Price] – used to compute the 
sales amount by summing the quantity multiplied by the net price. This calculation is widely used in 
reports, along with other columns like Quantity, Net Price, and Unit Cost. 

This is the code of the Sales Amount measure: 

--------------------------- 
-- Measure in Sales table 
--------------------------- 
Sales Amount = SUMX ( Sales, Sales[Quantity] * Sales[Net Price] ) 

The question is rather simple. Is it better to perform the multiplication at query time – using an iterator 
like we are doing in Sales Amount – or is it better to create a calculated column containing the 
multiplication result and then sum the calculated column at query time? 

Creating a calculated column comes with several consequences: 

• The model size grows because the calculated column is stored in RAM;

• The model processing time increases because the calculated column is computed sequentially at
process time.

The price to pay for a calculated column is a larger model that takes more time to process. At the same 
time, we expect that computing the value in advance will reduce the execution time of queries. To make 
an educated decision, we evaluate the different impacts. 

Let us start by creating the calculated column: 

------------------------------------- 
-- Calculated column in Sales table 
------------------------------------- 
Line Amount (USD) = Sales[Quantity] * Sales[Net Price] 

We must evaluate the impact on both process time and size. Before measuring the time required to 
compute this calculated column, we note that the column will be used only in DAX calculations: therefore, 
we do not need to store its hierarchy. The hierarchy would consume processing time and space in 
memory, so we clear the AvailableInMDX property of the column. 

Once we have created the column in the model, we process the model, triggering this way the 
calculation of only this column. We cannot easily measure the time required to compute only this new 
column. Nonetheless, because right after its creation the column is the only unprocessed entity in the 
entire model, measuring the default process of the model provides a good approximation of the actual 
cost. 

Before starting the process default, we open SQL Server Profiler and trace the Progress Report events. 
Then, we run a refresh by using the following TMSL script in SQL Server Management Studio: 

{ 
  "refresh": { 
    "type": "automatic", 
    "objects": [ 

 { 
"database": "Contoso 100M x 10" 

      } 
    ] 
  } 
} 

Because we monitor the process and the profiler gathers all progress report events, we should avoid 
using the server for anything else until the process finishes. The model is quite large; the Sales table 
contains 1.4B rows, so we do not need to measure timings in an extremely precise way. 

Once the processing is finished, we can look at the events to discover the time required to compute 
the column. 

The processing started at 17:18:40 and finished at 17:20:22, meaning that the calculated column took 
1 minutes and 42 seconds to compute. It is also worth noting that the server used a single core during 
the entire calculation. This is a known limitation of AS, which computes calculated columns sequentially. 
The time spent to compute a calculated column cannot be reduced by increasing the number of cores. 
Nonetheless, we also know that if the result is good we can compute this column as a calculated column 
in the view that feeds the table – so that instead of being a calculated column, the Line Amount column 
becomes a regular imported column. 

We perform the tests with a calculated column because we do not want to process 1.4B rows every 
time we run a test. The results with a calculated column are slightly imprecise, but the time required to 
gather them is so tiny that it is worth proceeding this way. 

We can now use the VertiPaq Analyzer information to assess the size of the column. 

The entire Sales table uses 27GB of RAM, with the Line Amount (USD) column alone using 3.4GB. In 
other words, the calculated column alone uses around 14% of the total size of the database. The impact 
is quite significant in terms of RAM. 

Now that the column is in place, we can measure the benefits – if any – in terms of performance. 
Because we are interested in an overall measure, we do not slice by any column and just compute the 
sales amount with either the new calculated column or the iteration. We author the following query and 
compute first the Calc Column row alone, and then the Measure row, alone again: 

EVALUATE 
SUMMARIZECOLUMNS ( 

 "Calc column", SUM ( Sales[Line Amount (USD)] ) 
) 

Here is the result with the calculated column. 

The important number is the SE CPU, which is not dependent on the number of cores. Indeed, we use 
a server with a very large number of cores, reaching a very high degree of parallelism. SE CPU is a better 
indicator of the raw power consumed to produce the result when evaluating performance. 

Next, the result using the measure, therefore performing the multiplication for every row: 

EVALUATE 
SUMMARIZECOLUMNS ( 

 "Measure", SUMX ( Sales, Sales[Quantity] * Sales[Net Price] ) 
) 

As expected, the calculated column is beneficial to performance. The calculated column takes 1,781 
milliseconds to aggregate its values, whereas the multiplication at query time produces the same result 
using 4,016 milliseconds of SE CPU. 

The question is whether the benefit is worth the increase in size. Before making up our mind, we want 
to look at further considerations. The test focuses strictly on the Line Amount calculated column. 
Nonetheless, we have other calculations following the same pattern: cost and margin. Therefore, we create 
two more calculated columns and measures: 

------------------------------------- 
-- Calculated column in Sales table 
------------------------------------- 
Line Cost (USD) = Sales[Quantity] * Sales[Unit Cost] 

------------------------------------- 
-- Calculated column in Sales table 
------------------------------------- 
Line Margin (USD) = Sales[Quantity] * ( Sales[Net Price] - Sales[Unit Cost] ) 

------------------------------------- 
-- Measure in Sales table 
------------------------------------- 
Cost Amount = SUMX ( Sales, Sales[Quantity] * Sales[Unit Cost] ) 

------------------------------------- 
-- Measure in Sales table 
------------------------------------- 
Margin Amount = SUMX ( Sales, Sales[Quantity] * ( Sales[Net Price] - Sales[Unit Cost] ) ) 

Then, we repeat the measurement process in a more complex query that uses the three columns and 
measures. First, we run this query to measure the time required to compute the values in the calculated 
columns: 

EVALUATE 
SUMMARIZECOLUMNS ( 

 "Sales Amount", SUM ( Sales[Line Amount (USD)] ), 
    "Cost Amount", SUM ( Sales[Line Cost (USD)] ), 
    "Margin Amount", SUM ( Sales[Line Margin (USD)] ) 
) 

The timing is in line with what we expect; the time required to compute three measures is around three 
times the time required to compute one measure. 

Then, we execute the following query to evaluate the time required to compute the values at query 
time: 

EVALUATE 
SUMMARIZECOLUMNS ( 

 "Sales Amount", [Sales Amount], 
    "Cost Amount", [Cost Amount], 
    "Margin Amount", [Margin Amount] 
) 

This time, the results are worse than expected, because of the larger time in the SE CPU. 

Indeed, gathering three columns using calculated columns increased the time required by a factor of 
around three. On the other hand, the same results computed by using measures are slower: 16 seconds 
against 4 seconds – four times more expensive. The reason is that Line Margin requires multiple columns 
in the same expression: Quantity, Net Price, and Unit Cost. Increasing the expression’s complexity makes 
the solution with the calculated column preferable. 

Nonetheless, we should also evaluate the increase in terms of RAM. Here is the VertiPaq Analyzer 
report with the three calculated columns. 

The three columns use 9.6GB of RAM, massively increasing the model size. 

Though it is undoubtedly true that a calculated column brings good benefits to the model 
performance, at the same time, the price in terms of RAM consumption seems excessive. Creating one 
calculated column for each calculation would make the model too large. 

Here is where a conclusion is hard to make without considering further details. If this were a general-
purpose model, it would not make much sense to consolidate all the possible calculations in calculated 
columns, because the price in terms of space used would definitely be too large. On the other hand, if 
you have a complex calculation that involves multiple columns and intricate logic, and this calculation is 
at the core of most of your reports, then a calculated column to consolidate those results would make 
sense. 

In the particular case of Contoso, with Sales Amount, Cost Amount, and Margin Amount, it is likely 
better to avoid the calculated columns to save memory, thus paying the price of working with slower 
queries – even though in smaller models the difference might not measurable. 

As you have seen, a simple decision about whether to compute a calculated column or not requires 
quite a few considerations and measurements. Besides, just knowing the details of why we made a choice 
already has a lot of value. We did not choose randomly. We performed measurements and made decisions 
based on solid numbers. If at any point we need to change our mind, we already know the consequences 
of making any choice. 

Optimizing composite models 

The first optimization example we saw in this chapter was related to a simple DAX optimization. For the 
sake of simplicity, we used an example using a regular import model, therefore handling only xmSQL 
queries to the VertiPaq engine. Nonetheless, it is important to note that optimizing DAX requires a deep 
understanding of the entire model architecture. 

As an example, we see how the composite model feature affects the process of DAX optimization. We 
use as an example the dynamic segmentation pattern. We will experience bad performance: the reason is 
not the DAX code, but rather the interaction between the local and remote engines. 

This example aims to demonstrate that optimizing DAX requires more than just DAX knowledge. The 
same DAX code performs very differently if executed in a regular VertiPaq model, or in a composite model. 
In order to optimize DAX for a composite model, you need to have a profound understanding of the 
architecture and how the queries are split between the remote and the local servers. 

We use a variation of the pattern available at Dynamic segmentation – DAX Patterns 
(https://www.daxpatterns.com/dynamic-segmentation/) We implement dynamic clustering on a model 
hosted in the Power BI Service by extending it with a composite model. Based on a configuration table, 
we want to dynamically cluster the customers by Sales Amount. The following is the configuration table 
created in the composite model. 

We create a measure using a slightly simplified version of the code published on 
www.daxpatterns.com: 

https://www.daxpatterns.com/dynamic-segmentation/
https://www.daxpatterns.com/dynamic-segmentation/
http://www.daxpatterns.com/

--------------------------- 
-- Measure in Sales table 
--------------------------- 
Customer in segment := 
SUMX ( 

 'Sales Segment', 
    VAR MinSale = 'Sales Segment'[MinValue] 

 VAR MaxSale = 'Sales Segment'[MaxValue] 
 VAR CustInSeg = 

FILTER ( 
 Customer, 
VAR CustSales = [Sales Amount] 
 RETURN 

 CustSales > MinSale && CustSales <= MaxSale 
) 

 RETURN 
 COUNTROWS ( CustInSeg ) 

) 

The measure mixes the Sales Segment table stored in the local VertiPaq model with the Customer table 
stored in the remote model. It works just fine: you can generate the following report showing the number 
of customers in each segment, sliced by year. 

Although the numbers are correct, there is a performance issue. The measure would work very well in 
a regular VertiPaq model, but it is extremely slow in a composite model. The query that populates the 
matrix (somewhat simplified here) is the following: 

-- 
--  Query executed on the local model 
-- 
EVALUATE 
SUMMARIZECOLUMNS ( 

 ROLLUPADDISSUBTOTAL ( 'Sales Segment'[Segment], "IsGrandTotalRowTotal" ), 
    ROLLUPADDISSUBTOTAL ( 'Date'[Year], "IsGrandTotalColumnTotal" ), 
    "Customer_in_segment", 'Sales Segment'[Customer in segment] 
) 

Looking at the server timings, we discover that this simple query runs in around three seconds on a 
rather small model with a few thousand sales: 

Further analysis shows that the complexity is hidden in the storage engine DAX queries sent to the 
remote server. If we analyze those queries, we find a first query that is already suspicious: 

-- 
--  DAX DirectQuery query executed on the remote model 
-- 
EVALUATE 
SELECTCOLUMNS ( 

 'Customer', 
 "__RN", blank(), 
 "'Customer'[CustomerKey]", 'Customer'[CustomerKey], 
 "'Customer'[Gender]", 'Customer'[Gender], 
 "'Customer'[Name]", 'Customer'[Name], 
 "'Customer'[Address]", 'Customer'[Address], 
 "'Customer'[City]", 'Customer'[City], 
 "'Customer'[State Code]", 'Customer'[State Code], 
 "'Customer'[State]", 'Customer'[State], 
 "'Customer'[Zip Code]", 'Customer'[Zip Code], 
 "'Customer'[Country Code]", 'Customer'[Country Code], 
 "'Customer'[Country]", 'Customer'[Country], 
 "'Customer'[Continent]", 'Customer'[Continent], 
 "'Customer'[Birthday]", 'Customer'[Birthday], 
 "'Customer'[Age]", 'Customer'[Age] 

) 

This query retrieves all the columns of the Customer table. In our example, there are just a few thousand 
customers. In a more realistic scenario, you could have millions of customers. Therefore, the query is 
simple, but the datacache returned is potentially huge. Besides, it seems like neither the measure nor the 
report require any of those columns. 

The next two queries are even worse, and they provide an explanation about why the previous query 
retrieved the entire Customer table. The first query retrieves the sales amount for each customer and year, 
using as a filter the entire customer table. In other words, the customer table that the local DAX engine 
has retrieved from the remote model is used as a filter in a subsequent query sent to the remote engine: 

-- 
--  DAX DirectQuery query executed on the remote model 

-- 
DEFINE 

 VAR _Var0 = 
VALUES ( 'Date'[Year] ) 

 VAR _Var1 = 
SUMMARIZE ( 

 'Customer', 
'Customer'[CustomerKey], 
'Customer'[Gender], 
'Customer'[Name], 
'Customer'[Address], 
'Customer'[City], 
'Customer'[State Code], 
'Customer'[State], 
'Customer'[Zip Code], 
'Customer'[Country Code], 
'Customer'[Country], 
'Customer'[Continent], 
'Customer'[Birthday], 
'Customer'[Age] 

 ) 
 VAR _Var2 = {

      ( 1212508, "Male", "David Puente", "189 Koontz Lane", "Los Angeles", "CA", 
"California", "90017", "US", "United States", "North America", DT"1992-1-4", 29 ), 

 ( 1200226, "Male", "William Gaughan", "4384 Euclid Avenue", "Guadalupe", "CA", 
"California", "93434", "US", "United States", "North America", DT"1942-6-26", 78 ), 

      -- 
 --  Several thousands of rows here, one for each customer 
 -- 
 ( 1200334, "Male", "Micheal Boyers", "655 Carriage Court", "Los Angeles", "CA", 

"California", "90017", "US", "United States", "North America", DT"1989-9-30", 31 ), 
      ( 1201225, "Male", "John Lally", "3886 Kerry Way", "Irvine", "CA", 

"California", "92614", "US", "United States", "North America", DT"1996-2-6", 24 ), 
      ( 395073, "Male", "Michael Vandermark", "386 Dufferin Street", "Toronto", "ON", 

"Ontario", "M6H 4B6", "CA", "Canada", "North America", DT"1983-1-3", 38 ) 
 } 

EVALUATE 
GROUPCROSSAPPLYTABLE ( 

 'Date'[Year], 
    _Var0, 
    _Var1, 
    "L1", 

GROUPCROSSAPPLY ( 
 'Customer'[CustomerKey], 
'Customer'[Gender], 
'Customer'[Name], 
 'Customer'[Address], 
'Customer'[City], 
'Customer'[State Code], 
'Customer'[State], 
'Customer'[Zip Code], 
'Customer'[Country Code], 
'Customer'[Country], 
'Customer'[Continent], 
'Customer'[Birthday], 

 'Customer'[Age], 
KEEPFILTERS ( 

 TREATAS ( 
 _Var2, 
'Customer'[CustomerKey], 
'Customer'[Gender], 
'Customer'[Name], 
'Customer'[Address], 
'Customer'[City], 
'Customer'[State Code], 
'Customer'[State], 
'Customer'[Zip Code], 
'Customer'[Country Code], 
'Customer'[Country], 
'Customer'[Continent], 
'Customer'[Birthday], 
'Customer'[Age] 

 ) 
 ), 
 "__Agg0", [Sales Amount] 

 ) 
) 

Not only is this query huge, but so is its resulting data cache. Indeed, the query result is – again – the 
entire Customer table along with the Sales Amount result. Therefore, this is a large query with a large 
result… Definitely an issue. 

A third query is very similar to the previous one, where the only noticeable difference is that the year 
is no longer part of the group by section. Still, the query includes the entire Customer table, and the 
resulting datacache contains the entire Customer table. 

In other words, the entire Customer table is passed back and forth between the local and the remote 
servers multiple times. The number of customers in each segment is later computed in the formula engine 
based on the data retrieved from the remote server. 

Now that the reason why the query is slow is somewhat clear, it is time to fix the problem. One of the 
few golden rules of DAX is to never use a table to apply a filter. A filter over a table requires scanning the 
entire table. The VertiPaq engine contains several optimizations and patterns that reduce the number of 
columns scanned. However, the DirectQuery over AS engine has a reduced set of optimizations. Therefore, 
Tabular ends up scanning the entire table, just because this is what we required in the DAX code. 

More specifically, the problem is the reference to Customer in the FILTER part of the measure. It turns 
out that we do not need to count the rows in the Customer table; it is enough to retrieve only the 
Customer[CustomerKey] column to obtain a semantically equivalent measure: 

--------------------------- 
-- Measure in Sales table 
--------------------------- 
Customer in segment = 
SUMX ( 

 'Sales Segment', 
    VAR MinSale = 'Sales Segment'[MinValue] 

 VAR MaxSale = 'Sales Segment'[MaxValue] 
 VAR CustInSeg = 

FILTER ( 
 DISTINCT ( Customer[CustomerKey] ), 
VAR CustSales = [Sales Amount] 
 RETURN 

 CustSales > MinSale && CustSales <= MaxSale 
) 

 RETURN 
 COUNTROWS ( CustInSeg ) 

) 

By operating this small change in the formula, we obtain the very same result, but the DAX queries 
sent to the remote engine are much simpler, hence the timings. 

The DirectQuery query executed is way simpler than before: 

-- 
--  DAX DirectQuery query executed on the remote model 
-- 
DEFINE 

 VAR _Var0 = VALUES ( 'Date'[Year] ) 
 VAR _Var1 = VALUES ( 'Customer'[CustomerKey] ) 
 VAR _Var2 = VALUES ( 'Customer'[CustomerKey] ) 

EVALUATE 
GROUPCROSSAPPLYTABLE ( 

 'Date'[Year], 
    _Var0, 
    _Var1, 
    "L1", 

GROUPCROSSAPPLY ( 
 'Customer'[CustomerKey], 
KEEPFILTERS ( _Var2 ), 
"__Agg0", [Sales Amount] 

) 
) 

This query retrieves the value of Sales Amount for each customer and year, exactly as the previous 
query did. The big difference is that we no longer have the ridiculous filter – instead, we obtain a smaller 
query, and the result set contains only the customer key, the year, and the sales amount. Consequently, 
the resulting data cache is smaller. 

As you have seen, we updated the DAX code specifically for a composite model in this example. A 
measure that works just fine on a regular VertiPaq model might show performance issues with a 
composite model. A similar scenario applies with DirectQuery models: VertiPaq and a relational database 
like SQL Server are very different engines, so the DAX code optimization depends on the storage engine(s) 
used. 

Conclusions 

The goal of this chapter was not to teach any technique. We optimized DAX code, processing, and models 
without describing the rationale behind each decision in detail. The important part was to show that 
optimizing a Tabular model requires knowing several details and using many different tools. Without 
proper knowledge about the internals, it is nearly impossible to produce efficient code. A good DAX 
developer should master each of these techniques and tools. 

Starting from the next chapters, we go into the details and build the knowledge needed to optimize 
your Tabular models. It will take some time before we can put the learning into practice. Even though at 
some point you might feel that all the details we provide are not so relevant… Take a deep breath and 
continue ingesting knowledge. As soon as we start to optimize, you will see that each detail is helpful. 

